
1 WSimply

WSimply is an extension of Simply, which lets the user deal with Weighted CSP
and tackle several meta-constraints, besides its previous CSP solving capabili-
ties.

In the following subsections both are introduced, first Simply in section 1.1
and then WSimply in section 1.2.

1.1 Simply architecture

The basic architecture of Simply is shown in figure 1. The input of the Simply
system are CSP instances, written in its own language. Those instances are
reformulated into SMT instances, which are fed into an SMT solver. If the
SMT solver is able to find a solution, then this solution is translated back to a
solution of the original CSP instance. Moreover, several optimization algorithms
can be run on top of the SMT solver (which is supposed to solve only decision
problems).

Figura 1: Basic architecture of Simply.

A Simply instance is formed by a model file, defining the problem, and an
optional data file, giving specific data of the problem instance.

1.1.1 Model File

The model file always starts with the keyword Problem, followed by the charac-
ter “:” and a string naming the problem. After that several sections are used
to model the problem:

• Data definition

• Domain definition

• Variable definition

• Predefinition of Functions and Constraints (optional)

• Constraint posting



• Optimization (optional)

〈simply program〉 ::= Problem:〈id〉 〈data〉 〈domains〉 〈variables〉
〈user defined functions〉? 〈constraints〉
〈optimitzation〉?

As a modelling example we show the classical n-queens problem:

Problem:queens

Data
int n=8;

Domains
Dom rows=[1..n];

Variables
IntVar q[n]::rows;

Constraints

AllDifferent([q[i]|i in [1..n]]);

Forall(i in [1..n-1]) {
Forall(j in [i+1..n]) {

q[i]-q[j]<>j-i;
q[j]-q[i]<>j-i;

};
};

Data definition

Inside the model file, the data definition section is where the problem con-
stants are defined, and optionally a value is assigned to them. Otherwise its
suposed that the value will be taken from the data file.

Example:

Data
int n;

〈data〉 ::= Data 〈data def〉∗
〈data def〉 ::= int 〈id〉 = 〈arithm exp〉 ;

| bool 〈id〉 = 〈formula〉 ;
| (bool | int) 〈array id〉 = 〈list〉 ;

Domain definition

Inside the model file, the domain definition section is where the integer
variable domains are defined.

Example:



Domains
Dom rows=[1..n];

〈domains〉 ::= Domains 〈domain def〉∗
〈domain def〉 ::= Dom 〈id〉 = 〈list〉 ;

Variable definition

Inside the model file, the variable definition section is where variables and
indexed variables (integer or boolean) are defined. In case of integer variables
a domain is assigned to them.

Example:

Variables
IntVar q[n]::rows;

〈variables〉 ::= Variables 〈variable def〉∗
〈variable def〉 ::= IntVar 〈array id〉 ( , 〈array id〉)∗ :: (〈id〉 | int) ;

| BoolVar 〈array id〉 ( , 〈array id〉)∗ ;
〈array id〉 ::= 〈id〉

| 〈id〉[ 〈arithm exp〉 ( , 〈arithm exp〉)∗ ]

User Functions and Constraints definition

Inside the model file, we also can predefine functions or constraints. For
instance, in the queens modelling example we can define the diagonal attack as
the following constraint:

DefConstraint non_diagonal_attack(int t,int i,int j)
{

t[i]-t[j]<>j-i;
t[j]-t[i]<>j-i;

}

and then in the constraints posting we only have to refer this constraint
using its name and parameters. Similarly we can define functions that return a
boolean or integer value.:

〈user defined functions〉 ::= Functions 〈user defined functions def〉∗
〈user defined functions def〉 ::= 〈def constraint〉

| 〈def function〉 ;
〈def constraint〉 ::= DefConstraint 〈id〉 〈dfc params〉 { 〈sentence〉+ }
〈def function〉 ::= Function (int | bool) 〈id〉 〈dfc params〉 〈sentence〉
〈dfc params〉 ::= ( (int | bool) 〈id〉 (, (int | bool) 〈id〉)∗ )

Constraint posting
Inside the model file, the constraints posting section is where the problem is

modelled. Constraints posting starts with the keyword Constraints and then
one or more sentences. A sentence is a simple constraint or a comprehension



list that posts a conjunction of constraints (generated by the list). A simple
constraint is a formula or a gobal constraint.

A formula can be a negation of another formula, a binary boolean operation
(or, and, etc), a binary relational operation (<, >, =, etc), a reference to a user
defined constraint, a If Then Else that is evaluated at solving time, an state-
ment (forall, exists or if-then-else) that are evaluated during compilation time,
a formula between parenteses, a boolean variable or True and False constants.

A global constraint can be:

• Sum(List,Value). This constraint enforces equality between Value and
the sum of all elements of List. When List is empty, Value is enforced
to be zero.

• Count(List,Value,N). This constraint states equality between N and the
number of occurrences of Value in List. When List is empty, N is en-
forced to be zero.

• AtMost(List,Value,N). This constraint states that the number of occur-
rences of Value in List must be less or equal than N. When List is empty,
N is enforced to be zero.

• AtLeast(List,Value,N). This constraint states that the number of oc-
currences of Value in List must be less or equal than N. When List is
empty, N is enforced to be zero.

• ExactlyOne(List). This constraint states that the number of satisfied
constraints in List must be exactly one.

• AllDifferent(List) requires all the elements of List to be different.

Following the previous example, we used the global constraint AllDifferent,
two statements: two nested Foralls, and we changed the basic constraints for
user defined reference.

Constraints

AllDifferent([q[i]|i in [1..n]]);

Forall(i in [1..n-1]) {
Forall(j in [i+1..n]) {

non_diagonal_attack(q,i,j);
};

};

〈constraints〉 ::= Constraints 〈sentence〉+
〈sentence〉 ::= 〈constraint〉 ;

| 〈list〉 ;



〈constraint〉 ::= 〈formula〉
| 〈global constraint〉

〈formula〉 ::= Not 〈formula〉
| 〈formula〉 〈bool operator〉 〈formula〉
| 〈arithm exp〉 〈relational operator〉 〈arithm exp〉
| 〈user defined references〉
| If Then Else (〈formula〉) { 〈sentence〉+} { 〈sentence〉+}
| 〈statement〉
| ( 〈formula〉 )
| 〈array id〉
| True
| False

〈user defined references〉 ::= 〈id〉 〈dfc param〉

〈relational operator〉 ::= = | <> | < | > | =< | >=

〈bool operator〉 ::= And | Or | Xor | Iff | Implies

〈global constraint〉 ::= AllDifferent( 〈list〉 )
| Sum( 〈list〉 , 〈arithm exp〉 )
| Count( 〈list〉 , (〈arithm exp〉|〈formula〉) , 〈arithm exp〉 )
| AtLeast( 〈list〉 , 〈arithm exp〉 )
| AtMost( 〈list〉 , 〈arithm exp〉 )
| ExactlyOne( 〈list〉 )

〈arithm exp〉 ::= 〈numeral〉
| 〈array id〉
| 〈arithm exp〉 〈arithm operator〉 〈arithm exp〉
| ( 〈arithm exp〉 )
| Abs( 〈arithm exp〉 )
| 〈user defined references〉
| length( 〈arithm exp〉 )
| max( 〈list〉 )
| min( 〈list〉 )
| sum1( 〈list〉 )

We want to remark that the arithmetic expressions length, max and min are
evaluated in compilation time only, therefore the elements in 〈list〉 also have to
be evaluated in compilation time.

Note that the arithmetic expression sum1 is exactly equal to the operator +,
but it is useful when we the number of elements to sum is undefined. It can be
used in compilation time (when all the elements could be evaluated in compila-
tion time) or can be used for generating of constraint.

There are three types of statements in Simply, that are evaluated during
compilation time:

• Forall: generates a conjunction of sentences



• Exists: generates a disjunction of sentences

• If-Then-Else: needs a condition, and generates the sentences of the Then
if the condition is evaluted to true (in compilation time) otherwise gener-
ates the sentences of the Else.

It is important to notice the difference between the If-Then-Else statement
and the If Then Else constraint, which, as we have said, is not evaluated in
compilation time. Consider, for instance, the following example: If (i < 4)
Then { m[i] <> m[i+1]; } Else { m[i] <> m[i-1]; m[i] = m[i-2]; }. Con-
dition i < 4 is evaluated in compilation time. Therefore, the variable i cannot
be a “constraint” variable, i.e., it must be a constant or a “local” variable, e.g.,
an index of a Forall statement.

〈statement〉 ::= 〈if then else〉
| 〈forall〉
| 〈exists〉

〈if then else〉 ::= If ( 〈formula〉 ) Then { 〈sentence〉+ }
| If ( 〈formula〉 ) Then { 〈sentence〉+ } Else { 〈sentence〉+ }

〈forall〉 ::= Forall( (〈id〉 in 〈list〉 )+ ) { 〈sentence〉+ }
〈exists〉 ::= Exists( (〈id〉 in 〈list〉 )+ ) { 〈sentence〉+ }

〈arithm operator〉 ::= + | - | * | Div | Mod

There are two types of lists in Simply:

• Extensional lists: by directly enumerating elements and ranges

• Intensional lists: using comprehension lists.

〈list〉 ::= [ 〈list element〉 ( , 〈list element〉)∗ ]
| [ (〈arithm exp〉|〈formula〉) | 〈var restrict〉( , 〈var restrict〉)∗]

〈list element〉 ::= 〈arithm exp〉
| 〈range〉

〈var restrict〉 ::= 〈id〉 in 〈list〉
| 〈formula〉

〈range〉 ::= 〈arithm exp〉 .. 〈arithm exp〉

〈id〉 ::= non empty sequence of letters, digits and , not starting with digit
〈numeral〉 ::= non empty sequence of digits
〈comments〉 ::= % anything else until end of line

| \ * anything else * \

Optimization
Inside the model file, the user can optionally specify to solve the problem

maximizing or minimizing an objective variable. Otherwise only the satisfiabil-
ity of the problem will be checked.

〈optimitzation〉 ::= Minimize 〈id〉 ;
| Maximize 〈id〉 ;

Figura 2: Basic architecture and solving process of WSimply.

1.1.2 Data file

Inside the data file, we can only assign values to data. The data definition is
inside the model file, so in the data file we cannot determine the type or the
number of dimensions of the data elements.

〈data file〉 ::= 〈data file def〉∗
〈data file def〉 ::= 〈id〉 = 〈arithm exp〉 ;

| 〈id〉 = 〈formula〉 ;
| 〈id〉 = 〈list〉 ;

1.2 WSimply architecture

WSimply is an extension of Simply with the following improvements:

• Soft-constraints, extending the constraint definition to deal with Weighted
CSP

• Meta-constraints, giving more control over the soft-constraints violations

Figure 2 shows the basic architecture and solving process of WSimply. WSimply
reformulates the input instance into the suitable format for the solving proce-
dures. We have four reformulations: (R1) from a WCSP instance with meta-
constraints into a WCSP instance (without meta-constraints), (R2) from a WC-
SP instance into a COP instance, (R3) from a WCSP instance into a WSMT
instance and (R4) from a COP instance into a WSMT instance.

Once the problem has been properly reformulated, we can apply two different
solving approaches: WSMT Solving (S1) or Optimisation SMT Solving (S2).

1.2.1 Model file

Inside de model file, the meta-constraint definition section is added as optional
after the constraints posting.



〈simply program〉 ::= Program:〈id〉 〈data〉 〈domains〉 〈variables〉
〈user defined functions〉? 〈constraints〉
〈meta constraints〉? 〈optimitzation〉?

To deal with the soft-constraints the constraints posting is extended allowing
weights to the sentences using an @ and the expression of the weight. Also is
allowed to define a label to the constraint in order to reference that constraint in
other constraints or in the meta-constraints. More details of the label definition
can be found in section 1.2.2.

〈constraints〉 ::= Constraints 〈sentence〉+
〈sentence〉 ::= 〈extended sentence〉 ;

| 〈list〉 ;
〈extended sentence〉 ::= 〈label def〉? 〈constraint〉 〈weight〉?
〈label def〉 ::= # 〈array id〉 :
〈weight〉 ::= @ { 〈weight value〉 }

| @ 〈weight value〉
〈weight value〉 ::= 〈arithm exp〉

|

1. Priority. The user may have some preferences on the soft constraints
to violate. For instance, if we have an activity to perform and worker 1
doesn’t want to perform it while worker 2 should not perform it, then it is
better to violate the first constraint than the second. It would be useful to
free the user of deciding the exact value of the weight of each constraint.

To this end, we allow the use of undefined weights denoted as “ ”:

#label : (constraint) @ { };

The value of this undefined weight is computed at compile time accord-
ing to the meta-constraints that refer to the label. This simplifies the
modelling of the problem, since the user does not need to compute any
concrete weight.

WSimply provides the following meta-constraints related to priority:

• samePriority(List), where List is a list of labels of weighted con-
straints. With this meta-constraint we are stating that the con-
straints referred to in the List are soft constraints with the same
priority.

• priority(List), where List is a list of labels of weighted constraints.
With this meta-constraint we are stating that the constraint corre-
sponding to the i-th label in List has higher priority than the con-
straint corresponding to the (i + 1)-th label. In other words, it must
be more costly to violate the i-th constraint than the (i + 1)-th.

• priority(label1,label2,n), with n > 1, defines how many times it
is worse to violate the constraint corresponding to label1 than to



violate the constraint corresponding to label2. That is, if weight1
and weight2 denote the weights associated with label1 and label2,
respectively, we are stating that weight1 ≥ weight2 ∗ n.

• multiLevel(ListOfLists), where ListOfLists is a list of lists of labels.
This meta-constraint states that the cost of falsifying each of the
constraints referred to by the labels in the i-th list is greater than
the cost of falsifying the constraints referred to by the labels in the
rest of the lists (from the i + 1-th list to the last) all together. For
example,

multiLevel([[A,B,C],[D,E,F],[G,H,I]]);

states that the cost of falsifying each of the constraints (denoted by)
A, B and C is greater than the cost of falsifying D, E, F, G, H, and
I together and, at the same time, the cost of falsifying each of the
constraints D, E and F is greater than the cost of falsifying G, H, and
I together.

2. Homogeneity. The user may wish there to be some homogeneity in the
amount of violation of disjoint groups of constraints. For instance, for the
sake of fairness, the number of violated preferences of nurses should be as
homogeneous as possible. WSimply provides the following meta-constraints
related to homogeneity:

• atLeast(List,p), where List is a list of labels of soft constraints and
p is a positive integer in 1..100. This meta-constraint ensures that
the percentage of constraints denoted by the labels in List that are
satisfied is at least p.

• homogeneousAbsoluteWeight(ListOfLists,v), where ListOfLists is
a list of lists of labels of soft constraints and v is a positive integer.
This meta-constraint ensures that, for each pair of lists in ListOfLists,
the difference between the cost of the violated constraints in the two
lists is at most v. For example, given

homogeneousAbsoluteWeight([[A,B,C],[D,E,F,G]],10);

if the weights of constraints A, B and C are 5, 10 and 15 respectively,
and constraints A and B are violated and constraint C is satisfied, then
the cost of the violated constraints in [D,E,F,G] must be between 5
and 25.

• homogeneousAbsoluteNumber(ListOfLists,v). Same as above, but
where the maximum difference v is between the number of violated
constraints.

• homogeneousPercentWeight(ListOfLists,p), where ListOfLists is a
list of lists of labels of soft constraints and p is a positive integer in



1..100. This meta-constraint is analogous to homogeneousAbsoluteWeight,
but where the maximum difference p is between the percentage in the
cost of the violated constraints (with respect to the cost of all the
constraints) in each list.

• homogeneousPercentNumber(ListOfLists,p). Same as above, but
where the maximum difference p is between the percentage in the
number of violated constraints.

We remark that the homogeneousAbsoluteWeight and homogeneousPercentWeight
meta-constraints are not allowed to refer to any constraint with undefined
weight. This is because, constraints with undefined weight are referenced
by priority meta-constraints, and their weight is determined at compile
time accordingly to those priority meta-constraints, independently from
other constraints.
Hence, since the homogeneousAbsoluteWeight and homogeneousPercentWeight
meta-constraints also constrain the weight of the referenced constraints, if
they were allowed to reference constraints with undefined weights, this
could lead to incompleteness. We also remark that these two meta-
constraints can not refer to constraints whose weights contain decision
variables Nevertheless, note that we can homogenise constraints using
homogeneousAbsoluteNumber and homogeneousPercentNumber meta-constraints,
without worrying about the weights.

3. Dependence. Particular configurations of violations may entail the ne-
cessity to satisfy other constraints, that is, if a soft constraint is violated
then another soft constraint must not, or a new constraint must be satis-
fied. For instance, working the first or the last turn of the day is penalized
but, if somebody works in the last turn of one day, then he cannot work
in the first turn the next day. This could be succinctly stated as follows:

#A: not_last_turn_day_1 @ w1;
#B: not_first_turn_day_2 @ w2;
(Not A) Implies B;

stating that, if the soft constraint A is violated, then the soft constraint B
becomes hard.

〈meta constraints〉 ::= 〈meta constraint def〉 ;
〈meta constraint def〉 ::= priority( 〈list〉 )

| priority( 〈array id〉 , 〈array id〉 , 〈arithm exp〉 )
| samePriority( 〈list〉 )
| multiLevel( 〈list of lists〉 )
| atLeast( 〈list〉 , 〈arithm exp〉 )
| homogeneousAbsoluteWeight( 〈list of lists〉 , 〈arithm exp〉 )
| homogeneousPercentWeight( 〈list of lists〉 , 〈arithm exp〉 )
| homogeneousAbsoluteNumber( 〈list of lists〉 , 〈arithm exp〉 )
| homogeneousPercenteNumber( 〈list of lists〉 , 〈arithm exp〉 )



1.2.2 Label definition

The label definition could be an identifier or an array identifier. The first case
only refers to one constraint, while the second one refers to several constraints.
Note that such constraints must be inside a Forall statement, this is because
the index of the label will be taken from the Forall index variable. Furthermore
such variable must represent a continuous range starting from 1.

In the following example, we can see an example of an array label called
ARLAB with index variable i which ranges from 1 to n elems-1.

Forall(i in [1..n_elems-1]) {
#ARLAB[i]:(ele[i] > ele[i+1])@1;

};


